Domain Specific Languages with F# 

Sourcecode: www.paul-abraham.com/DSLPolynomial.zip
Domain Specific Language (DSL) is a programming language which is applied on a specific area. Domain Specific Languages have habitually very small syntax and offers limited expressions, consequently it is very easy to learn and the user can focus on solving a problem rather than learning a fully fledged language. Pattern matching capabilities of F# and tools for parsing (fsyacc.exe) & lexing(fslex.exe) allow you create own  DSL effortlessly. Martin Fowler has written a profound article in this subject (http://martinfowler.com/dslwip/UsingDsls.html)
Evaluating polynomial Expression                                                                                                                            I like to show you, how to evaluate a polynomial Expression. There are lots of mathematical formulas in a simulation application which may fine tuned for special cases. An expression interpreter can offer you great flexibly than hard coded expressions. Here are the main steps to create an interpreter.                                                                                                                                                                                                Step 1: Define an Abstract Syntax Tree (AST). The abstract syntax tree defines the syntactic structure of the language      
 #light

module PolynomialAst

// Definition of Abstract Syntax Tree

type Expr = 

  | Ident of string  
  | Pow of Expr * Expr 

  | Val of System.Double 
  | Multi of Expr * Expr 
  | Div of Expr * Expr

  | Minus of Expr * Expr

  | Plus of Expr * Expr 
  | Sin of Expr       // not a strict polynomial element , it is used here to demonstrate unary expressions     
Here are some examples:

f(x)=4 -> Val(4), f(x)=x+4 -> Plus(Ident ‘x’ ,Val 4.0), f(x)=x^2-4->Minus(Pow(Ident ‘x’,Val 2),Val 4.0)
Step 2: In the step, we will create a lexer with the help of F# tool “fslex.exe”. The lexer will use defined regular expression to extract elementary pieces from source code by mapping expressions(lexeme) into Tokens(http://en.wikipedia.org/wiki/Lexical_analysis). We will now specify lexer(Polylex.fsl) to create the lexer.
	{

open System

open PolyPars

open Lexing

}

let alp=['a'-'z' 'A'-'Z']+

let digit = ['0'-'9']

let whitespace = [' ' '\t' ]

let newline = ('\n' | '\r' '\n')

rule token = parse

| whitespace
{ token lexbuf }

| newline { token lexbuf }

| "("               { LPAREN }

| ")"               { RPAREN }

| "*"               { MULTI }

| "/"               { DIV }

| "Sin"             { SIN}

| "+"               { PLUS }

| "^"               { POW}

| "-"               { MINUS }

| alp
            { ID (lexeme lexbuf) }

| ['-']?digit+('.'digit+)?   

                    { FLOAT (Double.Parse(lexeme lexbuf)) }

| eof   { EOF }


Use the following command to create the lexer 
C:\Programme\FSharp-1.9.6.2\bin\fslex.exe G:\BlogArticles\DSL\DSLPolynomial\DSLPolynomial\PolyLex.fsl

Step 3: Lexer will break the source code and categorize in tokens and parser will take the tokens and will build an abstract syntax tree. We will specify lexer and  we will  use F# tool “fsc.exe” to create the lexer. 
	{%

open PolynomialAst

%}

%start Expression

%token <string> ID

%token <System.Double> FLOAT

%token LPAREN RPAREN POW EOF MULTI DIV PLUS MINUS SIN

%type < PolynomialAst.Expr > Expression

// Associativity and Precedence-Lowest precedence come first

%left PLUS

%left MINUS

%left MULTI

%left DIV

%left POW

%left SIN

%%

Expression: ID { Ident $1  }

    | FLOAT {  Val $1  }

    | SIN Expression { Sin($2)}

    | LPAREN Expression RPAREN {  $2  }

    | Expression POW Expression {  Pow($1, $3)  }

    | Expression MULTI Expression {  Multi($1, $3)  }

    | Expression DIV Expression {  Div($1, $3)  }

    | Expression PLUS Expression {  Plus($1, $3)  }

    | Expression MINUS Expression {  Minus($1, $3)  }


Use following command to create  the parser:

C:\Programme\FSharp-1.9.6.2\bin\fsyacc.exe G:\BlogArticles\DSL\DSLPolynomial\DSLPolynomial\PolyPars.fsy
Step 4: We will get AST expression from lexer and we have to interpret it  and we pass a dictionary which contains values for variables:
#light

open System.Collections.Generic

open PolynomialAst

open System.Threading

/// interprets Abstract Syntax Tree

let rec interpretAST(inExpr:Expr,varValues:Dictionary<string,float>)=

        match inExpr with

        |Ident(x)->varValues.[x]

        |Val(v)-> v

        |Pow(e1,e2)-> System.Math.Pow(interpretAST(e1,varValues),interpretAST(e2,varValues))

        |Multi(e1,e2)->interpretAST(e1,varValues)*interpretAST(e2,varValues)

        |Div(e1,e2)->interpretAST(e1,varValues)/interpretAST(e2,varValues)

        |Plus(e1,e2)->interpretAST(e1,varValues)+interpretAST(e2,varValues)

        |Minus(e1,e2)->interpretAST(e1,varValues)-interpretAST(e2,varValues)

        |Sin(e1)-> System.Math.Sin(interpretAST(e1,varValues))

 // set culture Info to en-US otherwise flot numbers for german os will be wrong

Thread.CurrentThread.CurrentCulture<- new System.Globalization.CultureInfo("en-US")  

let expString="x^2 - 4"

let  varMap:Dictionary<string,float> = new Dictionary<string,float>()

varMap.Add("x",2.0)

let lexBuff=Lexing.from_string(expString)

let exp=PolyPars.Expression PolyLex.token lexBuff

printfn "The AST Expression %A" exp

let answer = interpretAST(exp,varMap)

printfn "The answer is %f" answer
enjoy(
